

## MLC Lab Visit - Lab 05 - Maple

Mth 355 (a.k.a. Mth 399) Feb 05, 2003 Maple 7  
Bent E. Petersen

[petersen@math.orst.edu](mailto:petersen@math.orst.edu)

There are 6 problems below. Problem solutions are due Feb 12, 2003. Email your solutions to me as Maple worksheet attachments. Your worksheet must execute correctly for full credit.

### **Linear Recurrence Equations - Generating Functions**

```
[ > restart;
```

We considered already, a little bit, the generating function of a recurrence relation in lab 04. This section therefore is repetition.

The Maple function `rsolve()` is used to solve linear recurrence relations. It can also be used to find the generating function of the solution to a linear recurrence relations. Consider the example (based on the Fibonacci sequence):

```
[ > eqn1:=a(n)=a(n-1)+a(n-2)+(-1)^n*n;
```

$$eqn1 := a(n) = a(n - 1) + a(n - 2) + (-1)^n n$$

```
[ > init1:=a(0)=0,a(1)=1;
```

$$init1 := a(0) = 0, a(1) = 1$$

```
[ > g1:=rsolve({eqn1,init1},a,'genfunc'(z));
```

$$g1 := -\frac{\left(2\frac{1}{1+z} - \frac{z}{(1+z)^2}\right)z^2 + z}{-1 + z + z^2}$$

Note the single quotes surrounding `genfunc` in the command above.

Once we have the generating function we can find the  $a(n)$  by computing the Taylor series with center at the origin.

```
[ > taylor(g1,z=0,21);
```

$$z + 3z^2 + z^3 + 8z^4 + 4z^5 + 18z^6 + 15z^7 + 41z^8 + 47z^9 + 98z^{10} + 134z^{11} + 244z^{12} + 365z^{13} \\ + 623z^{14} + 973z^{15} + 1612z^{16} + 2568z^{17} + 4198z^{18} + 6747z^{19} + 10965z^{20} + O(z^{21})$$

Of course Maple will also find a procedure to compute a(n):

```
> fn1:=rsolve({eqn1,init1},a,'makeproc');
```

Let's compute a(20) and compare it with the coefficient of z^20 in the Taylor series above:

```
> fn1(20); simplify(%); rationalize(%);
```

$$-\frac{2097152}{5} \frac{\sqrt{5}}{(1-\sqrt{5})^{21}} + \frac{2097152}{5} \frac{\sqrt{5}}{(1+\sqrt{5})^{21}} + 19$$
$$48224579994255360 \frac{1}{(-1+\sqrt{5})^{21} (1+\sqrt{5})^{21}}$$
$$10965$$

As you can see, Maple can be a little stubborn when it comes to giving us the answer in the form that we want, even when it is an integer. You just have to keep trying!

Note we do not have to rely on our eyesight to extract the coefficient of z^20 in the Taylor series - Maple will find it for us:

```
> subs(z=0,diff(g1,z$20))/20!;
```

10965

The expression z\$20 here generates a sequence of 20 z's - very handy!

```
> z$20;
```

z, z, z, z, z, z, z, z, z, z, z, z, z, z, z, z, z, z, z, z

```
>
```

## **- Graphs. Adjacency Matrix. Number of Paths of a Given Length.**

Graphs are supported in Maple by the networks package. Use the command ?networks to get help on the various procedures defined in the networks package.

```
> restart;
```

```
> with(networks): with(linalg):
```

Warning, the names charpoly and rank have been redefined

Warning, the protected names norm and trace have been redefined and unprotected

### Example 1

The `new()` command creates a graph with an empty set of vertices and an empty set of vertices.

```
> new(G1);
```

We can add vertices to our graph with the `addvertex()` command:


```
> addvertex([v1,v2,v3,v4],G1);  
v1, v2, v3, v4
```

We add edges by specifying their set or list of endpoints (in a list). If we do not provide names for the edges then the default names `e1`, `e2`, and so on, will be used. If we use sets of endpoints we obtain undirected edges, whereas if we use lists of endpoints we obtain directed edges.

```
> addedge([ {v1,v2}, {v2,v3}, {v3,v1}, {v3,v4} ], names=[edg1,edg2,edg3  
,edg4],G1);  
edg1, edg2, edg3, edg4
```


Maple can draw graphs though the routine provided is a bit primitive. We can not specify color nor line thickness.

```
> draw(G1);
```




We can exercise some control over the shape of the drawn graphs, but it is limited. Check Maple help for the details of the Concentric and Linear options.

```
> draw(Concentric([v1,v2,v4],[v3]),G1);
```



```
> draw(Linear([v1,v3],[v2,v4]),G1);
```



The adjacency matrix tells which vertices are joined by edges:  $M1[i,j] = 1$  if there is an edge joining  $v_i$  and  $v_j$ . Otherwise  $M1[i,j] = 0$ .

```
> M1:=adjacency(G1);
```

$$M1 := \begin{bmatrix} 0 & 1 & 1 & 0 \\ 1 & 0 & 1 & 0 \\ 1 & 1 & 0 & 1 \\ 0 & 0 & 1 & 0 \end{bmatrix}$$

### Example 2

The complete graph has an undirected edge joining each pair of vertices

```
> G2:=complete(5):
```


```
> vertices(G2);
```

{ 1, 2, 3, 4, 5 }

```
> edges(G2);
```

{ e1, e9, e10, e2, e3, e4, e5, e6, e7, e8 }

```
> draw(G2);
```



```
> M2:=adjacency(G2);
```

$$M2 := \begin{bmatrix} 0 & 1 & 1 & 1 & 1 \\ 1 & 0 & 1 & 1 & 1 \\ 1 & 1 & 0 & 1 & 1 \\ 1 & 1 & 1 & 0 & 1 \\ 1 & 1 & 1 & 1 & 0 \end{bmatrix}$$

### Example 3

A loop at a vertex  $v$ , that is, an edge joining  $v$  to itself, is specified as  $\{v\}$ . Unfortunately, the `draw()` command will choke on graphs containing loops, or will not display the loops.

```
> new(G3):  
> addvertex(v1,v2,G3);  
v1, v2  
> addedges([ {v1,v2}, {v2} ],G3);  
e1, e2
```

The `ends` command will list the vertices which form the ends of an edge:

```
> ends(e1,G3);  
{v1, v2}  
> ends(e2,G3);  
{v2}
```

The adjacency matrix in the presence of loops has 2's on the diagonal.

```
> M3:=adjacency(G3);  

$$M3 := \begin{bmatrix} 0 & 1 \\ 1 & 2 \end{bmatrix}$$

```

Sometimes one wants the loops to count only as 1, not as 2. We can take care of that easily:

```
> evalm(M3-diag(seq(M3[i,i]/2,i=1..coldim(M3))));  

$$\begin{bmatrix} 0 & 1 \\ 1 & 1 \end{bmatrix}$$


```

Make sure you understand how the previous command works. It is written so as to work for adjacency matrices of any size.

#### Example 4

The Petersen graph (no relation) is interesting:

```
> G4:=petersen();  
> draw(G4);
```



```
> M4:=adjacency(G4);
```

$$M4 := \begin{bmatrix} 0 & 1 & 0 & 0 & 1 & 1 & 0 & 0 & 0 & 0 \\ 1 & 0 & 1 & 0 & 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 1 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 & 1 & 0 & 1 & 0 & 0 & 0 \\ 1 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 1 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 1 \\ 0 & 0 & 0 & 1 & 0 & 1 & 0 & 1 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 1 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 & 1 & 0 & 1 \\ 0 & 0 & 1 & 0 & 0 & 1 & 0 & 0 & 1 & 0 \end{bmatrix}$$

The eigenvalues of the adjacency matrix can tell us quite a bit about a graph. In the case of the Petersen graph the eigenvalues are particularly simple.

```
> eigenvals(M4);
```

```
3, -2, -2, -2, -2, 1, 1, 1, 1, 1
```

### Example 5

Here's a graph with two directed edges:


```
> new(G5):
```

```
> addvertices([v1,v2,v3,v4,v5],G5);
```

```

                                v1, v2, v3, v4, v5
> addedges( [ {v1,v2}, [v1,v3], [v3,v5], {v1,v4}, {v4,v5}, {v2,v5}, {v3,v
4} ], G5);
                                e1, e2, e3, e4, e5, e6, e7
> draw(G5);

```



Unfortunately the draw command does not indicate direction. The adjacency matrix does however:

```
> M5:=adjacency(G5);
```

$$M5 := \begin{bmatrix} 0 & 1 & 1 & 1 & 0 \\ 1 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 1 & 1 \\ 1 & 0 & 1 & 0 & 1 \\ 0 & 1 & 0 & 1 & 0 \end{bmatrix}$$

Note we have a directed edge from v1 to v3 so  $M5[1,3] = 1$  but  $M5[3,1] = 0$ . Think of directed edges as one-way streets. Because of the directed edges  $M5$  is not symmetric. Nonetheless it has real eigenvalues:

```
> eigenvals(M5);
```

$$0, 0, -2, 1 + \sqrt{2}, 1 - \sqrt{2}$$

We may wonder if  $M5$  is diagonalizable, but then we run into a problem:

```
> is(M5,matrix);
```

*FAIL*

It appears the data structure returned by the `adjacency()` command is not recognized as a matrix by Maple! Fortunately we can force Maple to view it as a matrix by using the `evalm()` command. We can then compute the Jordan canonical form:

```
> jordan(evalm(M5));
```

$$\begin{bmatrix} 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & -2 & 0 & 0 \\ 0 & 0 & 0 & 1 - \sqrt{2} & 0 \\ 0 & 0 & 0 & 0 & 1 + \sqrt{2} \end{bmatrix}$$

We see that M5 is not diagonalizable.

### Example 6

Consider a simple graph, that is, a graph with no parallel edges, no loops, and no directed edges.

```
> new(G6):
```


```
> addvertices(v1,v2,v3,v4,G6);
```

*v1, v2, v3, v4*

```
> addedges([ {v1,v2}, {v1,v3}, {v1,v4}, {v2,v3}, {v3,v4} ],G6);
```

*e1, e2, e3, e4, e5*

```
> draw(G6);
```



```
> M6:=adjacency(G6);
```

$$M6 := \begin{bmatrix} 0 & 1 & 1 & 1 \\ 1 & 0 & 1 & 0 \\ 1 & 1 & 0 & 1 \\ 1 & 0 & 1 & 0 \end{bmatrix}$$

Now  $M6$  is a real symmetric matrix, so it has real eigenvalues and is diagonalizable.

```
> eigvals:=eigenvals(M6);
```

$$\text{eigvals} := 0, -1, \frac{1}{2} + \frac{1}{2}\sqrt{17}, \frac{1}{2} - \frac{1}{2}\sqrt{17}$$

Consider now a path in  $G6$  where we walk along successive edges. The length of a path is the number of edges in it. It takes only a moment of reflection to realize that the number of paths of length  $n$  from the  $i$ th to the  $j$ th vertex is the  $(i,j)$ th entry in the matrix  $M6^n$ . Since

```
> M6^20;
```

$$\begin{bmatrix} 45954543 & 35877321 & 45954542 & 35877321 \\ 35877321 & 28015882 & 35877321 & 28015882 \\ 45954542 & 35877321 & 45954543 & 35877321 \\ 35877321 & 28015882 & 35877321 & 28015882 \end{bmatrix}$$

we see there are 45,954,543 paths of length 20 from  $v1$  to  $v1$ . (Can you see all of them?)

This result is nice but not very convenient if we want to know how many paths of length  $n$ .

If we diagonalize  $M6$  we can obtain a convenient formula with a little bit of work.

```
> T6:=jordan(evalm(M6),'P6');
```

$$T6 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & \frac{1}{2} + \frac{1}{2}\sqrt{17} & 0 \\ 0 & 0 & 0 & \frac{1}{2} - \frac{1}{2}\sqrt{17} \end{bmatrix}$$

Note

```
> evalm( P6 &* T6 &* P6^(-1) - M6 ): simplify(%);
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

It follows that  $M6^n = P6 \&* T6^n \&* P6^{(-1)}$ . Let's introduce a symbolic version of  $T6^n$

```
> T6n:=diag(op(map(x->x^n,[seq(T6[i,i],i=1..coldim(T6))])));
```

$$T6n := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & (-1)^n & 0 & 0 \\ 0 & 0 & \left(\frac{1}{2} + \frac{1}{2}\sqrt{17}\right)^n & 0 \\ 0 & 0 & 0 & \left(\frac{1}{2} - \frac{1}{2}\sqrt{17}\right)^n \end{bmatrix}$$

(Make sure you understand the previous command.)

Then  $M6^n$  is given by

```
> M6n:= P6 &* T6n &* P6^(-1):
```

Now for the (1,1) entry we have:

```
> np:=evalm(M6n)[1,1];
```

$$np := \frac{1}{2}(-1)^n + \frac{1}{68} \left( \frac{1}{2} + \frac{1}{2} \sqrt{17} \right)^n (17 + \sqrt{17}) - \frac{1}{68} \left( \frac{1}{2} - \frac{1}{2} \sqrt{17} \right)^n (\sqrt{17} - 17)$$

This then is the number of paths of length  $n$  from  $v_1$  to  $v_1$ . Let's check it agrees with the above for  $n=20$ .

```
> subs(n=20,np): simplify(%);
45954543
```

Sure enough.

```
>
```

## - Problems

```
> restart;
```

### Problem 1

Find the generating function  $g$  for

```
> a:=n->sum(k^5,k=0..n);
```

$$a := n \rightarrow \sum_{k=0}^n k^5$$

**Hint:** Find a recurrence equation (and initial condition) satisfied by  $a(n)$  and then use `rsolve()`.

As a check the coefficient of  $z^{17}$  in the Taylor-Maclaurin expansion of  $g(z)$  is  $a(17) = 4767633$ .

### Problem 2

Find the generating function  $g$  for

```
> eqnp2:=b(n)=6*b(n-1)+3*4^n;
```

$$eqnp2 := b(n) = 6 b(n-1) + 3 \cdot 4^n$$

```
> initp2:=b(0)=2;
```

$$initp2 := b(0) = 2$$

### Problem 3

```
[ > unassign('a');
```

Consider the recurrence equation

```
[ > eqnp3:=a(n)=n*a(n-1);
```

*eqnp3 := a(n) = n a(n - 1)*

```
[ > initp3:=a(0)=1;
```

*initp3 := a(0) = 1*

The solution to this recurrence equation is pretty obvious, but if you ask Maple to compute the generating function, Maple chokes. Why?

#### Problem 4

Find the generating function *gp4* for the recurrence equation

```
[ > eqnp4:=b(n)=b(n-3)+4*n;
```

*eqnp4 := b(n) = b(n - 3) + 4 n*

```
[ > initp4:=b(0)=0,b(1)=0,b(2)=-1;
```

*initp4 := b(0) = 0, b(1) = 0, b(2) = -1*

Then find *b(20)*.

#### Problem 5

Write a procedure *taylorcoeff(g,n,z)* to return the the *n*th coefficient in the Taylor-Maclaurin expansion of an expression *g*. Write your procedure so the variable relative to which to expand is specified as the third argument.

As a check on your work

*taylorcoeff(1/(3\*z+w^3), 15, z);* returns  $-14348907 \frac{1}{w^{48}}$

*taylorcoeff(1/(3\*z+w^3), 15, w);* returns  $-\frac{1}{729} \frac{1}{z^6}$

```
[ > with(networks): with(linalg):
```

Warning, the names *charpoly* and *rank* have been redefined

Warning, the protected names *norm* and *trace* have been redefined and unprotected

#### Problem 6

Consider the 2 by 2 matrix

```
> Mp6:=matrix(2,2,[0,1,1,1]);
```

$$Mp6 := \begin{bmatrix} 0 & 1 \\ 1 & 1 \end{bmatrix}$$

Notice

```
> evalm(Mp6^2); evalm(Mp6^3); evalm(Mp6^4);
```

$$\begin{bmatrix} 1 & 1 \\ 1 & 2 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 2 \\ 2 & 3 \end{bmatrix}$$

$$\begin{bmatrix} 2 & 3 \\ 3 & 5 \end{bmatrix}$$

Clearly a simple induction will show that  $Mp6^n[2,2]$  is the  $n$ th Fibonacci number (assuming the 0th and first are 1). By diagonalizing  $Mp6$  deduce the Binet formula for the  $n$ th Fibonacci number. **Hint:** See example 6 in the section on graphs above.

Problems dealing with graphs are postponed until we have developed this topic in class.

```
>
```

```
>
```

```
>
```